

COMPUTER ALGEBRA SYSTEMS IN A MULTIVARIABLE CALCULUS COURSE AND CENTER OF GRAVITY PROBLEMS

Tilak DE ALWIS

Department of Mathematics
Southeastern Louisiana University
Hammond, LA 70402
USA
e-mail: talwis@selu.edu

ABSTRACT

In teaching a multivariable calculus course, two main difficulties facing the student are to draw surfaces in three-dimensions, and to setup and calculate tedious triple integrals. The added third dimension causes great difficulty even to the well prepared student who has successfully finished a two semester single variable calculus course. The student must now suddenly think in three-dimensions. Visualizing and drawing the corresponding three-dimensional surfaces pose a significant challenge to the novice. To alleviate the problem, the student and the instructor must resort to the modern technologies. Computer algebra systems (CAS) such as *Mathematica* and *Maple* are well equipped to handle such tasks. The paper has two goals: One goal is to demonstrate the usage of the CAS *Mathematica* to learn some standard topics of a multivariable calculus course, such as vectors, partial derivatives, graphing of three-dimensional objects, and multiple integrals. As the second, but most important goal of the paper, we will consider the special topic of the center of gravity of solid objects. This topic was chosen because it uses very many of the techniques learned in a multivariable calculus course. We will show how to use the CAS *Mathematica* to evaluate tedious triple integrals arising in calculating the center of gravity. *Mathematica* can also be used as a visualization tool to draw the graphs of three-dimensional solids under consideration. Usually a standard multivariable course only considers the center of gravity of fixed solids. However, with the aid of *Mathematica*, the students are in an ideal position to consider variable solids as well. Thus, the paper introduces the novel concept of the locus of the center of gravity of certain types of variable solids. The paper also illustrates several facets of A CAS in undergraduate education – the usage of a CAS as a computational tool, visualization tool, experimentation tool, and a conjecture-forming tool.

1. Computer Algebra Systems in a Multivariable Calculus Course

Some of the topics of a standard multivariable calculus course include vectors, partial derivatives, directional derivatives, surfaces in three-dimensions, extrema of functions of two variables, cylindrical and spherical coordinates, multiple integrals, and variable transformations (see [15] and [17]). The following diagram illustrates some of these different facets of a multivariable calculus course:




Figure 1.1 Different Facets of a Multivariable Calculus Course

The above diagram is not meant as to represent a complete exhaustive list of topics covered in a standard multivariable calculus course. In this section, we will discuss how to use the CAS *Mathematica* to learn some of the above topics. *Mathematica* is a general purpose CAS. It can be used as a numeric or symbolic computational device, a tool to draw two or three dimensional graphs, a visualization system to analyze data, or even as a multimedia studio to combine sounds and animation. The built-in programming language of *Mathematica* makes it an excellent tool to investigate mathematical or physical problems. Some good references on *Mathematica* are [2], [13], [18], and [19]. For the usage of *Mathematica* as a visualization tool, the reader can refer to [6], [8], [9], [11], and [12]. For the usage of *Mathematica* as a computational or a conjecture-forming tool, refer to [3], [4], [5], [7], and [10].

1.1 Vectors

(a) Dot Products:

The *Mathematica* command “**Dot**” can be used to calculate the dot product of any two vectors, two-dimensional or three-dimensional (see [19]).

Example 1.1 Find the dot product of the vectors $\mathbf{u} = \langle 1, 2, -3 \rangle$, and $\mathbf{v} = \langle -4, 1, 2 \rangle$.

Vectors are represented by objects in *Mathematica* called lists. For example, the vector \mathbf{u} above is given as the list $\{1, 2, -3\}$. To find the dot product of the two given vectors, use the following *Mathematica* command:

Input: $\mathbf{Dot}[\{1, 2, -3\}, \{-4, 1, 2\}]$

To execute the command, press “**Shift-Enter**”. The output is -8 . Therefore, $\mathbf{u} \cdot \mathbf{v} = -8$.

Note: Another way to input the dot product command in *Mathematica* is to use the “.” operation directly from the keyboard. For example, the following command “ $\{1, 2, -3\} \cdot \{-4, 1, 2\}$ ” yields the same result as before.

(b) Cross Products

The *Mathematica* command “**Cross**” can be used to compute the cross product of any two three-dimensional vectors (see [19]).

Example 1.2 Find the cross product of the vectors $\mathbf{u} = \langle 1, 2, -3 \rangle$ and $\mathbf{v} = \langle -4, 1, 2 \rangle$.

The following *Mathematica* command achieves the task:

Input: $\mathbf{Cross}[\{1, 2, -3\}, \{-4, 1, 2\}]$

Press “**Shift-Enter**” to execute the command. The output implies that $\mathbf{u} \times \mathbf{v} = \langle 7, 10, 9 \rangle$.

(c) Triple Scalar Products and Vector Identities

One important result on the dot and cross products is the following vector identity, where \mathbf{u} , \mathbf{v} , and \mathbf{w} are any three-dimensional vectors (see [15] and [17]):

$$\mathbf{u} \cdot (\mathbf{v} \times \mathbf{w}) = (\mathbf{u} \times \mathbf{v}) \cdot \mathbf{w} \quad (1.1)$$

Either side of the equation (1.1) is referred to as the triple scalar product of the vectors \mathbf{u} , \mathbf{v} , and \mathbf{w} .

Example 1.3 Use *Mathematica* to establish the vector identity (1.1).

The following *Mathematica* program perform the required task:

Input:

```
 $\mathbf{u} = \{\mathbf{u1}, \mathbf{u2}, \mathbf{u3}\};$   
 $\mathbf{v} = \{\mathbf{v1}, \mathbf{v2}, \mathbf{v3}\};$   
 $\mathbf{w} = \{\mathbf{w1}, \mathbf{w2}, \mathbf{w3}\};$ 
```

```

expr1=Dot[u,Cross[v,w]]
expr2=Dot[Cross[u,v],w]
Simplify[expr1-expr2]

```

Press “**Shift-Enter**” to execute the program. The first three lines of the program define the three vectors \mathbf{u} , \mathbf{v} , and \mathbf{w} . The next two lines will compute the left and right-hand sides of the equation (1.1). The last line of the program will compute and simplify the difference between the two sides of equation (1.1). As an output of the program, one can observe that this difference is zero, establishing equation (1.1).

The following example shows the famous connection between the triple scalar products and 3X3 determinants:

Example 1.4 Given that $\mathbf{u} = \langle u_1, u_2, u_3 \rangle$, $\mathbf{v} = \langle v_1, v_2, v_3 \rangle$, and $\mathbf{w} = \langle w_1, w_2, w_3 \rangle$, use *Mathematica* to show that the triple scalar product $\mathbf{u} \cdot (\mathbf{v} \times \mathbf{w})$ is given by the following 3X3 determinant:

$$\mathbf{u} \cdot (\mathbf{v} \times \mathbf{w}) = \begin{vmatrix} u_1 & u_2 & u_3 \\ v_1 & v_2 & v_3 \\ w_1 & w_2 & w_3 \end{vmatrix} \quad (1.2)$$

We can use the *Mathematica* command “**Det**” to evaluate the determinant of any square matrix (see [19]). Consider the following program:

Input:

```

u={u1,u2,u3};
v={v1,v2,v3};
w={w1,w2,w3};
expr1= Dot[u,Cross[v,w]]
expr2=Det[{{u1,u2,u3},{v1,v2,v3},{w1,w2,w3}}]
Simplify[expr1-expr2]

```

Press “**Shift-Enter**” to execute the program. The fourth and the fifth lines of the above program compute the left and right-hand sides of the equation (1.2) as expr1 and expr2, respectively. According to the output corresponding to the last line of the program, the difference between expr1 and expr2 is zero. This verifies the above equation (1.2). The importance of equation (1.2) is that it provides the volume of the parallelepiped formed by the vectors \mathbf{u} , \mathbf{v} , and \mathbf{w} with a common initial point (see [15] and [17]).

Example 1.5 Use *Mathematica* to calculate the equation of the plane passing through the point (x_0, y_0, z_0) with the normal vector $\langle a, b, c \rangle$.

Here is the idea: Let (x, y, z) be an arbitrary point on the required plane. Then $\mathbf{u} = \langle x - x_0, y - y_0, z - z_0 \rangle$ is a vector lying on the plane. Therefore, the vector \mathbf{u} must be perpendicular to the normal vector $\mathbf{n} = \langle a, b, c \rangle$, so $\mathbf{u} \cdot \mathbf{n} = 0$. Thus, the equation of the plane can be found by setting the dot product of the vectors \mathbf{u} and \mathbf{n} to be zero. So, consider the following *Mathematica* command:

Input: $\text{Dot}[\{x-x_0,y-y_0,z-z_0\},\{a,b,c\}] == 0$

Press “**Shift-Enter**” to execute the above command. The output confirms the following well-known equation of the required plane (see [15] and [17]):

$$a(x - x_0) + b(y - y_0) + c(z - z_0) = 0 \quad (1.3)$$

One can of course experiment with the command by assigning numerical values for x_0, y_0, z_0 , and a, b, c .

1.2 Partial Derivatives

The *Mathematica* command for differentiation, “**D**” can be used to compute all types of partial derivatives of multivariable functions. For example, the command “**D[f[x,y],{x,n}]**” computes the n th partial derivative of a function $f(x, y)$ with respect to x (see [19]).

Example 1.6 Given that $f(x, y) = x^2 + y^2 + x \sin(xy) + 5$, compute the partial derivatives $\partial f / \partial x$, $\partial f / \partial y$, $\partial^2 f / \partial x^2$, $\partial^2 f / \partial y^2$, $\partial^2 f / \partial x \partial y$, and $\partial^2 f / \partial y \partial x$.

The following commands perform the required task:

Input:

```
f[x_,y_]:= x^2 + y^2 + x*Sin[x*y] + 5
D[f[x,y],x]
D[f[x,y],y]
D[f[x,y],{x,2}]
D[f[x,y],{y,2}]
D[f[x,y],x,y]
D[f[x,y],y,x]
```

Press “**Shift-Enter**” to execute the above commands: The first line of the program defines the function $f(x, y)$. The other lines are self-explanatory. The last two lines compute the mixed partials $\partial^2 f / \partial x \partial y$ and $\partial^2 f / \partial y \partial x$. The six results are respectively $2x + xy \cos(xy) + \sin(xy)$, $2y + x^2 \cos(xy)$, $2 + 2y \cos(xy) - xy^2 \sin(xy)$, $2 - x^3 \sin(xy)$, $2x \cos(xy) - x^2 y \sin(xy)$, and $2x \cos(xy) - x^2 y \sin(xy)$.

Example 1.7 Use *Mathematica* to show that the function $f(x, y, z) = 1 / \sqrt{x^2 + y^2 + z^2}$

satisfies the following Laplace's Equation (see [15] and [17]):

$$\frac{\partial^2 f}{\partial x^2} + \frac{\partial^2 f}{\partial y^2} + \frac{\partial^2 f}{\partial z^2} = 0 \quad (1.4)$$

Here are the commands:

Input:

```
f[x_,y_,z_]:=1/Sqrt[x^2+y^2+z^2];
a=D[f[x,y,z],{x,2}];
b=D[f[x,y,z],{y,2}];
c=D[f[x,y,z],{z,2}];
Simplify[a+b+c]
```

The commands are executed by pressing “**Shift-Enter**”. An output of zero indicates that the given function satisfies the Laplace's Equation (1.4). A function such as $f(x, y, z) = 1/\sqrt{x^2 + y^2 + z^2}$ above, satisfying the Laplace's Equation (1.4) is called a harmonic function (see [15] and [17]). The first line of the above commands can be modified to discover other types of harmonic functions.

1.3 Three-Dimensional Graphs and Level Curves

(a) Three-Dimensional Graphs

Mathematica provides an excellent system to visualize three-dimensional graphs. Among other methods, one can use “**Plot3D**” or “**ParametricPlot3D**” commands to plot such graphs (see [13], [18], and [19]).

Example 1.8 Use *Mathematica* to graph the hyperboloid $f(x, y) = x^2 - y^2$.

Here is the “**Plot3D**” command to perform the task:

```
Plot3D[x^2-y^2,{x,-7,7},{y,-7,7},PlotPoints->25]
```

Output:




Figure 1.2 The graph of the saddle $f(x, y) = x^2 - y^2$

One can use the *Mathematica* command “**ViewPoint**” to look at the surface from different camera angles (see [13], [18], and [19]). Execute the following command to look at the surface from the point (2, 2.5, 0.1):

Input: `Plot3D[x^2-y^2,{x,-7,7},{y,-7,7},PlotPoints->25,ViewPoint->{2, 2.5, 0.1}]`

Output:




Figure 1.3 The saddle from a different viewpoint

In this paper, we have used *Mathematica* version 3.0. However, note that the *Mathematica* version 4.0 also enables the user to rotate 3D graphs in real time using the “<<**RealTime3D**” command (see [13]).

Example 1.9 Use *Mathematica* to graph the sphere $x^2 + y^2 + z^2 = 4$.

Here it is more convenient to use the “**ParametricPlot3D**” command of *Mathematica*. One can parametrize the sphere using the spherical coordinates (see [15] and [17]): For example, $x = 2\cos\theta \sin\phi$, $y = 2\sin\theta \sin\phi$, $z = 2\cos\phi$ where $0 \leq \theta < 2\pi$ and $0 \leq \phi \leq \pi$ represents any point on the sphere. We use these coordinates with the “**ParametricPlot3D**” command:

Input: `ParametricPlot3D[{2Cos[theta]*Sin[phi],2Sin[theta]*Sin[phi],2Cos[phi]},
{theta, 0,2Pi},{phi,0,Pi}]`

Output:




Figure 1.4 The graph of the sphere $x^2 + y^2 + z^2 = 4$

One can restrict the parameters θ and ϕ to see an appropriate portion of the sphere. In other words, this provides a way of “cutting open” the sphere to see the inside: For example, execute the following *Mathematica* command to see what happens:

Input: `ParametricPlot3D[{2Cos[theta]*Sin[phi],2Sin[theta]*Sin[phi],2Cos[phi]},
{theta, 0, 3Pi/2},{phi, 0, Pi}]`

Output:




Figure 1.5 The sphere $x^2 + y^2 + z^2 = 4$ with an opening

(b) Level Curves of a Two Variable Function

In general, the level curves of a surface $z = f(x, y)$ are the curves in the XY -plane given by $f(x, y) = c$ where c is an arbitrary constant (see [15] and [17]). The “**ContourPlot**” command of *Mathematica* enables one to draw different level curves of a given surface (see [13] and [19]).

Example 1.10 Draw level curves of the surface given by $f(x, y) = x^2 + y^2$.

Input: `ContourPlot[x^2+y^2,{x,-5,5},{y,-5,5},Contours->10,PlotPoints->20]`

In the above, the option “**Contours->10**” will draw 10 contour lines corresponding to ten different heights (c -values):

Output:




Figure 1.6 The level curves of $f(x, y) = x^2 + y^2$

For the above example, the level curves are all circles. This is because any equation of the type $x^2 + y^2 = c$ where $c > 0$, represents a circle.

Example 1.11 Draw level curves for the surface $z = x^2 - y^2$ corresponding to the values $c = 1, -1,$ and 0 .

Mathematica can also be used to draw a specific level curve corresponding to a given c -value as this example requires. For example, to draw the level curve corresponding to $c = 1$, one can use the option “**Contours->{1}**”:

Input: `ContourPlot[x^2 - y^2, {x, -2, 2}, {y, -2, 2}, Contours ->{1}, ContourShading->False]`

Output:




Figure 1.7 The c -level curves of $z = x^2 - y^2$ for $c = 1$

The level curve corresponding to $c = 1$ is a hyperbola, because the equation $x^2 - y^2 = 1$ represents a hyperbola in the XY -plane. One can also plot all three level curves corresponding to the values $c = 1, -1, 0$ together in one diagram. In this case, use the option “**Contours->{1,-1,0}**”:

Input: `ContourPlot[x^2 - y^2, {x, -2, 2}, {y, -2, 2}, Contours ->{1, -1, 0}, ContourShading->False]`

Output:




Figure 1.8 The c -level curves of $z = x^2 - y^2$ for $c = 1, -1,$ and 0

It is true that *Mathematica* computes and plots with amazing efficiency, but unfortunately this might set a dangerous trend in students' minds. For example, some students might tend to believe that getting the final answer or the graph is the only objective, and might fail to see beyond this point. Therefore, it must be repeatedly stressed the importance of interpreting the answers obtained by a CAS. For instance, the students must be questioned as to why there are two straight lines in the above

level curve diagram. The reason is that the level curve corresponding to $c = 0$ is given by $x^2 - y^2 = 0$, which is equivalent to the pair of straight lines $y = \pm x$. If a CAS is not used with a very open and inquisitive mind, it can create permanent damage in the mathematical upbringing of the students!

1.4. Directional Derivatives, Gradients, and Tangent Planes

(a) Directional Derivatives and Gradients

Example 1.12 Find the directional derivative of the function $f(x, y) = x^2 + y^2 + \sin(xy)$ in the direction of the unit vector $\mathbf{u} = \langle 3/5, -4/5 \rangle$.

Note that the directional derivative of a function $z = f(x, y)$ in the direction of the unit vector $\mathbf{u} = \langle u_1, u_2 \rangle$ is given by (see [15] and [17])

$$D_{\mathbf{u}}f(x, y) = \nabla f(x, y) \cdot \mathbf{u} \quad (1.6)$$

where $\nabla f(x, y)$ is the gradient of the function $f(x, y)$ defined by

$$\nabla f(x, y) = \langle \partial f / \partial x, \partial f / \partial y \rangle \quad (1.7)$$

Thus, using the previously described *Mathematica* commands for the dot products and partial derivatives, one can compute the required directional derivative as follows:

Input:

```
u={3/5,-4/5};
f[x_,y_]:=x^2+y^2+Sin[x*y]
gradf={D[f[x,y],x],D[f[x,y],y]};
Dot[gradf,u]
```

Press “**Shift-Enter**” to obtain the required directional derivative as

$$D_{\mathbf{u}}f(x, y) = -(4/5)(2y + \cos(xy)) + (3/5)(2x + y\cos(xy)).$$

It must be noted that the two partial derivatives $\partial f / \partial x$ and $\partial f / \partial y$ are special cases of the directional derivative. For instance, when $\mathbf{u} = \langle 1, 0 \rangle$, the equations (1.6) and (1.7) imply that $D_{\mathbf{u}}f(x, y) = \langle \partial f / \partial x, \partial f / \partial y \rangle \cdot \langle 1, 0 \rangle = \partial f / \partial x$. The above program can be used to observe these facts as well.

Note: It must be noted that *Mathematica* has a built-in function “**Grad**” to compute the gradient of a function. However, before using this command, one must separately load the *Mathematica* vector analysis package by using the command “**<<Calculus`VectorAnalysis`**”. This package also has

other built-in commands such as “Div”, “Curl”, and “Laplacian” (see [19]). Also refer to Example 1.13 below.

(b) Tangent Plane to a surface

Consider a surface given by $f(x, y, z) = 0$. Suppose f is differentiable at the point (x_0, y_0, z_0) on the surface. Then a normal vector to the surface at (x_0, y_0, z_0) is given by the gradient $\nabla f(x_0, y_0, z_0)$ where

$$\nabla f(x, y, z) = \langle \partial f / \partial x, \partial f / \partial y, \partial f / \partial z \rangle \tag{1.8}$$

Here we are assuming that $\nabla f(x_0, y_0, z_0) \neq \mathbf{0}$. Therefore, under these conditions, the equation of the tangent plane to the surface at (x_0, y_0, z_0) is given by the following dot product equation (see [15] and [17]):

$$\nabla f(x_0, y_0, z_0) \cdot \langle x - x_0, y - y_0, z - z_0 \rangle = 0 \tag{1.9}$$

Example 1.13 Use *Mathematica* to find the equation of the tangent plane to the surface $z = 4 - x^2 - y^2$ at $(1, 1, 2)$. Also graph the surface and the tangent plane together.

The following *Mathematica* program uses the equation (1.9) to compute the equation of the required tangent plane. The *Mathematica* vector analysis package was used to calculate the gradient conveniently as mentioned before.

Input:

```
<<Calculus`VectorAnalysis`
f[x_,y_]:=4-x^2-y^2
phi[x_,y_,z_]:=f[x,y]-z
{x0,y0,z0}={1,1,f[1,1]};
tgpl=Simplify[ Dot[Grad[phi[x,y,z],Cartesian[x,y,z]]/. {x->x0,y->y0,z->z0},
               {x-x0,y-y0,z-z0}]]
p1=Plot3D[f[x,y],{x,-3,3},{y,-3,3},DisplayFunction->Identity];
p2=Graphics3D[{RGBColor[1,0,0],PointSize[1/60],Point[{1,1,f[1,1]}]}]
p3=Plot3D[tgpl+z,{x,0,2},{y,0,2},DisplayFunction->Identity]
Show[{p1,p3,p2},DisplayFunction->$DisplayFunction]
```

As the output, one obtains the equation of the tangent plane as $6 - 2x - 2y - z = 0$. The program also produces the following graphs of the surface together with the tangent plane at the point $(1, 1, 2)$:




Figure 1.9 The tangent plane to the surface $z = 4 - x^2 - y^2$ at $(1,1,2)$

1.5 Multiple Integrals

The *Mathematica* command “**Integrate**” can be used to compute many multiple integrals (see [19]):

Example 1.14 Calculate the triple Integral $\int_{x=-1}^{x=1} \int_{y=2x}^{y=x^2} \int_{z=x-y}^{z=x+y} xy^2 z(2xy + y^2 z) dz dy dx$.

Input: **Integrate**[$x*y^2*z(2x*y+y^2*z)$,{ $x,-1,1$ },{ $y,2x,x^2$ },{ $z,x-y,x+y$ }]

Press “**Shift-Enter**” to obtain the output as $-256/45$.

In the next sections of the paper, we will consider an important topic of a multivariable calculus course, namely the center of gravity of solids. This topic uses several fundamental concepts of a multivariable calculus course, such as partial derivatives, three-dimensional graphing, multiple integrals, variable transformations and Jacobians, normal lines and tangent planes. Thus, the center of gravity problems provide us an excellent opportunity to present the usage of a CAS in a multivariable calculus course.

2. The Center of Gravity of Three-Dimensional Solids

Consider the continuous function $z = f(x, y)$ defined on a region R in the XY -plane. We will assume that $f(x, y) \geq 0$ for any $(x, y) \in R$. Let S be the solid under the graph of f , directly sitting above the plane region R . Then the center of gravity $G(\bar{x}, \bar{y})$ of the solid S is given by (see [1], [14], [15], [16], and [17]).

$$\bar{x} = I_1 / I \quad (2.1)$$

$$\bar{y} = I_2 / I \quad (2.2)$$

$$\bar{z} = I_3 / I \quad (2.3)$$

where the integrals I_1, I_2, I_3 and I are defined by

$$I_1 = \iiint_D x dV \quad (2.4)$$

$$I_2 = \iiint_D y dV \quad (2.5)$$

$$I_3 = \iiint_D z dV \quad (2.6)$$

$$I = \iiint_D dV \quad (2.7)$$

In the above integrals (2.4)-(2.7), D denotes the three-dimensional region defined by the solid S .

Mathematica can be used to calculate the above integrals efficiently. The following examples illustrate how to use *Mathematica* as a computational and a visualization tool to understand the basic center of gravity problems:

Example 2.1 Find the center of gravity of the solid bounded by the graphs of $z = x^2 + y^2 + 1$, $x = -1$, $x = 1$, $y = -1$, $y = 1$, and $z = 0$.

As discussed in section 1 of the paper, the following “**Plot3D**” command of *Mathematica* can be used to visualize our solid.

Input: `Plot3D[x^2+y^2+1, {x,-1,1}, {y,-1,1}, PlotPoints->40, ViewPoint ->{2.081, -2.552, 0.779}]`

Output:




Figure 2.1 The solid bounded above by $z = x^2 + y^2 + 1$

One can now setup and calculate the integrals (2.4)-(2.7). For example,

$$\begin{aligned} I_1 &= \int_{x=-1}^1 \int_{y=-1}^1 \int_{z=0}^{x^2+y^2+1} x dz dy dx = \int_{x=-1}^1 \int_{y=-1}^1 x(x^2 + y^2 + 1) dy dx = \int_{x=-1}^1 \left[x^3 y + x \frac{y^3}{3} + xy \right]_{y=-1}^1 dx \\ &= \int_{x=-1}^1 \left(2x^3 + \frac{2}{3}x + 2x \right) dx = \left[\frac{x^4}{2} + \frac{x^2}{3} + x^2 \right]_{x=-1}^1 = 0 \end{aligned}$$

As the problem gets more complicated, the above types of triple integrals become more tedious to do by hand. Thus, A CAS becomes very helpful with the calculation. As discussed in section 1, one can use the “**Integrate**” command of *Mathematica* to compute the above integral (see [13] and [19]):

Input:

Integrate[x,{x,-1,1},{y,-1,1},{z,0,x^2+y^2+1}]

The output is zero, which means $I_1 = 0$. Similarly, one can use *Mathematica* to obtain $I_2 = 0$, $I_3 = 266/45$, and $I = 20/3$. Then the equations (2.1)-(2.3) imply that $\bar{x} = 0$, $\bar{y} = 0$, and $\bar{z} = 133/150$. This means that the center of gravity of the solid is given by $G(0,0,133/150)$. At this point, the students must be questioned as to why the first two coordinates of the center of gravity are zero. The reason is that our solid is symmetric about the z -axis, so its center of gravity must lie on the z -axis, which implies $\bar{x} = 0$ and $\bar{y} = 0$.

Example 2.2 Find the center of gravity of the solid bounded by the cylinder $x^2 + y^2 = 4$, the planes $x + y + z = 5$, and $z = 0$.

The “**ParametricPlot3D**” command can be used to graph the cylinder, while the “**Plot3D**” command can be used to graph the plane. The first two lines of the following program graphs these two objects, and then suppresses the output using the option “**DisplayFunction->Identity**”. The final line of the program combines the cylinder and the plane using the “**Show**” command, and displays back the combined output using the option “**DisplayFunction->\$DisplayFunction**” (see [13] and [19]).

Program 2.1

```
p1=ParametricPlot3D[{2Cos[theta],2Sin[theta],z},{theta,0,2Pi},{z,0,8},
PlotPoints->40,DisplayFunction->Identity];
p2=Plot3D[(5-x-y),{x,-2,2},{y,-2,2},PlotPoints->40,DisplayFunction->Identity];
Show[{p1,p2},DisplayFunction->$DisplayFunction, ViewPoint->{1.416, -1.191, 2.833}]
```

The output is as follows:




Figure 2.2 The solid bounded by $x^2 + y^2 = 4$, $x + y + z = 5$, and $z = 0$

After visualizing the solid, one can now setup the integrals corresponding to equations (2.4)-(2.7). Note that the base of our solid is a circular region given by $x^2 + y^2 \leq 4$. Therefore, it is better to use cylindrical coordinates to evaluate our integrals. In other words, consider the variable transformation $x = r \cos \theta$, $y = r \sin \theta$, and $z = z$. The Jacobian J of the transformation is given by the following determinant of a 3X3 matrix (see [14], [15], [16], and [17]):

$$J = \begin{vmatrix} \partial x / \partial r & \partial x / \partial \theta & \partial x / \partial z \\ \partial y / \partial r & \partial y / \partial \theta & \partial y / \partial z \\ \partial z / \partial r & \partial z / \partial \theta & \partial z / \partial z \end{vmatrix} \quad (2.8)$$

The *Mathematica* command “**Det**” can be used to calculate the above 3X3 determinant:

```
Input:
x=r*Cos[theta];
y=r*Sin[theta];
j=Simplify[Det[
  {{D[x,r],D[x,theta],D[x,z]},{D[y,r],D[y,theta],D[y,z]},{D[z,r],D[z,theta],D[z,z]}]]]
```

The last input line above evaluates the Jacobian (2.8). According to the output, the Jacobian is given by $J = r$. Therefore, the integral (2.4) becomes $I_1 = \int_{r=0}^2 \int_{\theta=0}^{2\pi} \int_{z=0}^{5-r\cos\theta-r\sin\theta} (r \cos \theta)(r) dz d\theta dr$. The following *Mathematica* command evaluates it:

```
Input:
Integrate[r*Cos[theta]*r,{r,0,2},{theta,0,2Pi},{z,0,5-r*Cos[theta]-r*Sin[theta}]]
```

The output is -4π , which means that $I_1 = -4\pi$. Similarly, it can be shown that $I_2 = -4\pi$, $I_3 = 54\pi$, and $I = 20\pi$. Then equations (2.1)-(2.3) imply that the center of gravity of

the solid is given by $G(-1/5, -1/5, 27/10)$. Observe that unlike the Example 2.1, the present solid is not symmetric around the z -axis, so the x and y -coordinates of its center of gravity are not zero.

In both of the above examples, we considered the center of gravity of fixed solids. But what will happen to the center of gravity if the solid is changed gradually? Let us consider the solid in Example 2.2 again. Recall that the upper boundary, or the roof, of this solid was given by the plane $x + y + z = 5$, and it passes through the point $(0, 0, 5)$ on the z -axis. The equation of this plane can be rewritten as $(x - 0)(1) + (y - 0)(1) + (z - 5)(1) = 0$. Therefore, it follows that the normal vector to the plane at the point $(0, 0, 5)$ is given by $\langle 1, 1, 1 \rangle$ (see [1], [14], [15], and [17]). One way of changing our solid is to change this normal vector gradually. One can imagine that as this normal vector is changing, the roof of the solid starts tilting around the fixed point $(0, 0, 5)$. As the solid changes, its center of gravity changes. It is of interest to track down this center of gravity in three-dimensional space. This leads to a series of interesting locus problems in three-dimensions (see [11] and [12]).

Before the CAS became popular, investigating the problems such as the center of gravity of variable solids was a nontrivial task. Such problems were never dealt with in an undergraduate curriculum because of the complexity of the calculations. However this situation has completely changed due to the wide availability of fast computers and CAS.

In the next section, we will utilize *Mathematica* to investigate the center of gravity of certain types of variable solids. As mentioned in the second to the last paragraph, our motivation comes from Example 2.2.

3. The Center of Gravity of a Class of Variable Solids

Consider the elliptic cylinder in three-dimensions given by the equation

$$x^2 / a^2 + y^2 / b^2 = 1 \tag{3.1}$$

where a and b are fixed positive constants. Consider the plane through the fixed point $(0, 0, c)$, $c > 0$, with variable normal vector $\langle s, t, 1 \rangle$ where s and t are real parameters. We will assume that c , s , and t are such that the plane will intersect the elliptic cylinder in the upper-half space $z > 0$. It is clear that the equation of the plane is given by (see [1], [15], and [17])

$$z = c - xs - yt \tag{3.2}$$

Let S_1 be the solid bounded by the elliptic cylinder (3.1), the plane (3.2), and the plane $z = 0$. See

the following figure:




Figure 3.1 The solid S_1 with the normal vector $\langle s, t, 1 \rangle$ at the point $(0, 0, c)$ on its roof

Let $G(\bar{x}, \bar{y}, \bar{z})$ be the center of gravity of the solid S_1 . As the parameters s and t change, the roof of the solid changes. Therefore, the center of gravity G of the solid S_1 changes. We want to investigate the locus of G in three-dimensions for changing s and t (see [11] and [12]). The following *Mathematica* program is written using the ideas described in Example 2.2. This program calculates the coordinates of the center of gravity G of the solid S_1 , and the locus of G for changing s and t . It then plots the graphs of the elliptic cylinder (3.1), and the locus of G in the same set of axes. Finally, the program makes an animation of G and the solid S_1 in the three-dimensional space.

Program 3.1

```

Clear[x,y,z,r,theta,a,b,c]
x=a*r*Cos[theta]; y=b*r*Sin[theta]; (* Defines the variable transformation *)
j=Simplify[Det[{{D[x,r],D[x,theta],D[x,z]}, {D[y,r],D[y,theta],D[y,z]},
               {D[z,r],D[z,theta],D[z,z]}]]; (* Calculates the Jacobian *)
ix=Integrate[j*x,{r,0,1},{theta,0,2Pi},{z,0,c-x*s-y*t};
iy=Integrate[j*y,{r,0,1},{theta,0,2Pi},{z,0,c-x*s-y*t};
iz=Integrate[j*z,{r,0,1},{theta,0,2Pi},{z,0,c-x*s-y*t};
i=Integrate[j*1,{r,0,1},{theta,0,2Pi},{z,0,c-x*s-y*t};
{x0,y0,z0}=Simplify[{ix/i,iy/i,iz/i}] (* Calculates G *)
Clear[x,y,z]
expr=z/.Solve[Eliminate[{x,y,z}=={x0,y0,z0},{s,t}],z][[1]] (* Calculates the locus of G *)
p1=ParametricPlot3D[Evaluate[{a*Cos[theta],b*Sin[theta],z} /.
  {a->1,b->2}],{theta,0,3Pi/2},{z,0,10},DisplayFunction->Identity] (* Defines the cylinder *)
p2=Plot3D[expr/{a->1,b->2,c->5},{x,-2,2},{y,-2,2},Mesh->False,
  DisplayFunction->Identity] (* Defines the locus *)
Show[{p1,p2},PlotRange->{{-2,2},{-2,2},{0,10}},
  DisplayFunction->$DisplayFunction] (*Plots the cylinder and the locus together)
p3:=Plot3D[c-x*s-y*t/{a->1,b->2,c->5,t->s/3+Sin[s]+Cos[s]},
  {x,-2,2},{y,-2,2},PlotRange->{0,10},DisplayFunction->Identity] (* Defines the roof *)
Do[Show[Graphics3D[ {PointSize[1/40],RGBColor[1,0,0],Point[{x0,y0,z0}]} /.
  {a->1,b->2,c->5, t->s/3+Sin[s]+Cos[s]}], p2,p1,p3, PlotRange->{0,10},
  DisplayFunction->$DisplayFunction], {s,-2,2,0.2}]

```

The program can be executed by pressing “**Shift-Enter**”. As the first output, one obtains the following coordinates of the center of gravity G of the solid S_1 :

$$G = \left(-\frac{a^2 s}{4c}, -\frac{b^2 t}{4c}, \frac{4c^2 + s^2 a^2 + t^2 b^2}{8c} \right) \quad (3.3)$$

As the second output, one obtains the equation of the locus of G , as given by equation below:

$$z = \frac{c}{2} + 2c \left(\frac{x^2}{a^2} + \frac{y^2}{b^2} \right) \quad (3.4)$$

In fact, one can obtain the equation (3.4) manually by eliminating the variables s and t from the three equations $x = -a^2 s / (4c)$, $y = -b^2 t / (4c)$, and $z = (4c^2 + s^2 a^2 + t^2 b^2) / (8c)$, which arise from equation (3.3). However, the Program 3.1 does this automatically, using the “**Eliminate**” command of *Mathematica* (see [19]). Note that the equation (3.4) represents an elliptic paraboloid opening up with z -intercept at $(0, 0, c/2)$. It is interesting to observe that the locus of the solid S_1 is an elliptic paraboloid.

The third output of the program plots the graphs of the elliptic cylinder (3.1), and the locus of G as given by equation (3.4) in the same set of axes, showing their relative positions:




Figure 3.2 The graphs of the elliptic cylinder and the locus of G

The final output of the program produces an animation of the solid S_1 with its tilting roof, along with its center of gravity. The animation can be run by grouping the graphic cells generated by the program into a single cell, and then by double clicking on this single cell. One can observe the different positions of the center of gravity G of the solid S_1 as a moving red dot. Note that the red dot always

lies on the elliptic paraboloid (3.4), which lies inside the elliptic cylinder (3.1). A few frames of the animation are given below:




Figure 3.3 An animation of the center of gravity G of the solid S_1

Example 3.1 Observe that the equation (3.3) providing the coordinates of the center of gravity G indeed agrees with the results of Example 2.2. Recall that in Example 2.2, the equation of the cylinder was $x^2 + y^2 = 4$, while the equation of the roof of the solid was $x + y + z = 5$. By comparing these with the equations (3.1) and (3.2), one finds that $a = 2$, $b = 2$, $c = 5$, $s = 1$, and $t = 1$. For these values, the equation (3.3) implies that $G = (-1/5, -1/5, 27/10)$, agreeing with the final result in Example 2.2.

One can summarize the findings of this section in the following Theorem (see [11] and [12]) :

Theorem 3.1 Consider the solid S_1 described in this section, bounded on the sides by the elliptic cylinder (3.1), bounded on the top by the plane (3.2), and bounded below by the plane $z = 0$. Then the center of gravity G of the solid S_1 is given by $G = (-a^2s/(4c), -b^2t/(4c), (4c^2 + s^2a^2 + t^2b^2)/(8c))$. For changing parameters s and t , the locus of G is an elliptic paraboloid with equation $z = c/2 + 2c(x^2/a^2 + y^2/b^2)$.

Proof. The student is encouraged to write a proof independent of *Mathematica*, using the methods discussed in Examples 2.1 and 2.2.

4. Conclusion

In this paper we observed how to use a CAS to understand several aspects of a multivariable calculus course, with the emphasis on the topic the center of gravity of a solid. This topic has inherent computational difficulties not just because of the third dimension, but also due to the tedious triple integrals and variable transformations. *Mathematica* can be used as a powerful computational tool to calculate those triple integrals involved. We also observed how to use *Mathematica* as a very effective visualization tool – not just static visualization, but also as a dynamic visualization tool. The

Mathematica programs we have used can also serve as a medium to experiment and a form conjectures on the center of gravity problems. Thus, the paper uncovers different facets of a CAS in undergraduate education. The paper also introduces a novel aspect of the center of gravity of solids, namely the study of the locus of the center of gravity of variable solids. This particular topic is not covered in calculus texts, traditional or otherwise. By introducing such nonstandard topics in a calculus course in conjunction with a CAS, one can take the undergraduate mathematics instruction to a new level. The paper uses *Mathematica* as the choice of CAS, but most of the ideas described here can be implemented by using other CAS such as Maple.

REFERENCES

- [1] Anton, H. (1999). *Calculus, Brief Edition*. New York, NY: John Wiley & Sons
- [2] Bahder, T. (1995). *Mathematica for Scientists and Engineers*. Redwood City, CA: Addison-Wesley.
- [3] de Alwis, T. (1993). *Mathematica and the Power Method*. *International Journal of Mathematics Education in Science and Technology*, 24(6), 813-824.
- [4] de Alwis, T. (1995). Projectile Motion with Arbitrary Resistance. *College Mathematics Journal*, 26(5), 361-367.
- [5] de Alwis, T. (1995). Families of Plane Curves Bounding a Constant Area. Proceedings of the First Asian Technology Conference in Mathematics, Nanyang Technological University, Singapore.
- [6] de Alwis, T. (1997). The Power of Animation in Visualizing Mathematics. Proceedings of the World Conference on Educational Multimedia and Hypermedia, Calgary, Canada.
- [7] de Alwis, T. (1997). Families of Plane Curves with a Constant Arc Length. *Innovation in Mathematics: Proceedings of the Second International Mathematica Symposium*, Rovaniemi Institute of Technology, Finland.
- [8] de Alwis, T. (1998). Normal Lines Drawn to a Parabola and Geometric Constructions. *Proceedings of the Third Asian Technology Conference in Mathematics*, University of Tsukuba, Japan.
- [9] de Alwis, T. (1999). Normal Lines Drawn to Ellipses and Elliptic Integrals. *Proceedings of the Third International Mathematica Symposium*, Research Institute of Symbolic Computation, Hagenburg, Austria.
- [10] de Alwis, T. (2000). Weighted Averaging Games and Difference Equations. *Mathematics and Computer Education Journal*, 34(1), 24-34.
- [11] de Alwis, T. (2001). The Locus of the Center of Gravity of Variable Solids. *Proceedings of the Third International Mathematica Symposium*, Tokyo Denki University, Chiba New Town Campus, Japan.
- [12] de Alwis, T. (2001). The Center of Gravity of Classes of Cylindrical Solids via a Computer Algebra System. *Proceedings of the Sixth Asian Technology Conference in Mathematics*, University of Melbourne, Australia.
- [13] Gray, T. and Glynn, J. (2000). *The Beginner's Guide to Mathematica, Version 4*. Cambridge, UK: Cambridge University Press.
- [14] Kreyszig, E. (1993). *Advanced Engineering Mathematics*. New York, NY: John Wiley & Sons.
- [15] Larson, H., Hostetler, R., and Edwards, B. (1994). *Calculus*. Lexington, MA: D.C. Heath.
- [16] O'Neil, P. (1991). *Advanced Engineering Mathematics*. Belmont, CA: Wadsworth Publishing Company.
- [17] Swokowski, E., Olinick, M., Pence D., and Cole, J. (1994). *Calculus*. Boston, MA: PWS Publishing Company.
- [18] Wickham-Jones, T. (1994). *Mathematica Graphics*. New York, NY: Springer Verlag.
- [19] Wolfram, S. (1996). *Mathematica Book, 3rd ed.* Cambridge, UK: Cambridge University Press.